NetzwerkDomainNamen ermitteln
Versionen: Access 2003, 2002/XP, 2000 und 97

 FRAGE:  Wir setzen Datenbanken ein, die im Netzwerk über verschiedene Server und Domains verteilt sind. Für die Vereinfachung von Supportanfragen möchten wir gerne im Formular anzeigen, in welcher Domain die aktuelle Datenbank betrieben wird. Können Sie uns einen Tipp geben, wie sich diese Information ermitteln lässt?
T. Carstensen

 ANTWORT:  Dazu nutzen Sie die API-Funktion "GetNetworkParams()", die unter anderem den Namen der Domain liefert, in der der aktuelle Anwender angemeldet ist. Geben Sie zunächst in einem neuen oder vorhandenen Modul die folgenden Deklarationen und Funktionen ein:

Private Const ERROR_SUCCESS = 0
Private Const MAX_DOMAIN_NAME_LEN = 128
Private Const MAX_HOSTNAME_LEN = 128
Private Const MAX_SCOPE_ID_LEN = 256 

Private Type IP_ADDRESS_STRING
    IpAddr(0 To 15) As Byte
End Type

Private Type IP_MASK_STRING
    IpMask(0 To 15) As Byte
End Type

Private Type IP_ADDR_STRING
    dwNext As Long
    IpAddress As IP_ADDRESS_STRING
    IpMask As IP_MASK_STRING
    dwContext As Long
End Type

Private Type FIXED_INFO
  HostName(0 To (MAX_HOSTNAME_LEN + 3)) As Byte
  DomainName(0 To (MAX_DOMAIN_NAME_LEN + 3)) As Byte
  CurrentDnsServer As IP_ADDR_STRING
  DnsServerList As IP_ADDR_STRING
  NodeType As Long
  ScopeId(0 To (MAX_SCOPE_ID_LEN + 3)) As Byte
  EnableRouting As Long
  EnableProxy As Long
  EnableDns As Long
End Type

Private Declare Function GetNetworkParams _
        Lib "iphlpapi.dll" _
        (pFixedInfo As Any, _
         pOutBufLen As Long) As Long 

Private Declare Sub CopyMemory _
        Lib "kernel32" Alias "RtlMoveMemory" _
        (hpvDest As Any, hpvSource As Any, _
         ByVal cbCopy As Long)

Function NetworkDomainName() As String
  Dim Buffer() As Byte
  Dim cbRequired As Long
  Dim FI As FIXED_INFO
  Dim strResult

  strResult = "???"
  GetNetworkParams 0&, cbRequired
  If cbRequired > 0 Then
    ReDim Buffer(0 To cbRequired - 1) As Byte
    If GetNetworkParams(Buffer(0), cbRequired) = _
                        ERROR_SUCCESS Then
      CopyMemory FI, ByVal VarPtr(Buffer(0)), LenB(FI)
      strResult = TrimNull(StrConv(FI.DomainName, _
                           vbUnicode))
      If strResult = "" Then 
        strResult = "Keine Domainanmeldung"
      End If
      NetworkDomainName = strResult
    End If 'If GetNetworkParams
  End If 'If cbRequired > 0

End Function

Legen Sie dann im Formular ein ungebundenes Textfeld an und geben Sie in der Eigenschaft "Steuerelementeinhalt" den folgenden Ausdruck ein:

=NetworkDomainName()

